



第3部

サブルーチン

日本原子力研究開発機構 田口 俊弘



第3部の講習内容

1. サブルーチンの書き方


2. 引数の使い方

3. 関数副プログラム

4. グローバル変数

付録：知っておくと便利な文法 3

1. サブルーチンの書き方





プログラムが長くなったら サブルーチンを使って分割しよう

サブルーチンとは

- 開始点と終了点を持ったメインプログラムとは独立したプログラム
- メインプログラムと異なり、動作中のプログラムから呼び出して利用する

サブルーチンの使い道

- 複数の場所で一連の同じ計算手順を使用するとき
- 方程式の解法や行列式の計算などのような定型処理をするとき
- 長いプログラムを分割してメンテナンスを容易にしたいとき

◎ メインプログラムを「メインルーチン」ということもあるので、サブルーチンとひとまとめにして呼ぶときは「ルーチン」という



サブルーチンの書式

☆サブルーチンは subroutine 文と end subroutine 文の間に書く

```
subroutine サブルーチン名          ! subroutine文
  implicit none
  real a,b
  integer i
  .....
end subroutine サブルーチン名      ! end subroutine文
```

- ★ ただし、データ受け渡しのために、subroutine文には「引数」を付けることができる（後で説明）
- ★ 「end subroutine サブルーチン名」は「end」だけでもエラーではないが、書いた方がよい

☆サブルーチンの内部はメインプログラムとほぼ同じ書式

- ★ 最初に implicit none を書く
- ★ 次に宣言文などの非実行文を書く
- ★ 非実行文の後に実行文を書く



サブルーチンには引数が付けられる

引数なしサブルーチン（サブルーチン名のみ）

```
subroutine サブルーチン名  
  implicit none  
  real a, b  
  integer i  
  .....  
  .....  
end subroutine サブルーチン名
```

引数ありサブルーチン

```
subroutine サブルーチン名(引数1, 引数2, ...)  
  implicit none  
  real a, b  
  integer i  
  .....  
  .....  
end subroutine サブルーチン名
```

- ★ 「引数」は、変数または配列名
- ★ 引数はサブルーチン内で宣言しなければならない



サブルーチンの利用手段 call 文

☆ サブルーチンだけのプログラムは動かない ☆

- サブルーチンは、他のルーチンから呼び出して初めて有効
- 呼び出しにはcall文を使う

引数なしサブルーチンの呼び出し

```
call サブルーチン名
```

引数ありサブルーチンの呼び出し

```
call サブルーチン名(数値1, 数値2, 数値3, ...)
```

- ★ 「数値」には「数式」を書いても良い（ただし注意あり！）
その場合には、数式の計算結果の数値がサブルーチンに渡される



プログラムを書いてみよう

3-A サブルーチンを使った四則計算

例題3-A

実数 x と y を与えると，その和・差・積・商の4個の値を計算して出力するサブルーチンを作成し，それを使って，いくつかの2個の実数を使ったときの結果を出力せよ



3-A サブルーチンを使った四則計算 の解答例

```
program answer3a
  implicit none
  real x, y
  x = 0.1;  y = 200
  call arithmetic(1.0, 2.0)
  call arithmetic(x, y)
  call arithmetic(2*x+1, x+y)
end program answer3a

subroutine arithmetic(x, y)
  implicit none
  real x, y
  print *, ' + - * / = ', x+y, x-y, x*y, x/y
end subroutine arithmetic
```

☆書き方のポイント☆

- ★ call 文で呼び出すときは、サブルーチンの引数として、定数、変数、計算式を書くことができる
- ★ ただし、数値型には気をつけなければならない（後で説明）
- ★ メインとサブで同じ名前の変数を宣言しているが、別物（後で説明）



サブルーチン実行の流れ

☆ サブルーチンを使ったときの実行の流れは以下の通り

- ① call 文を実行するとサブルーチンの実行開始点にジャンプ
- ② サブルーチンが終了するとcall文の次の行から実行を継続

例えば、次のプログラムの流れは. . .

```
program stest1
  implicit none
  real x, y
  x = 5.0
  y = 100.0
  call subr(x, y, 10)
  print *, x, y
end program stest1


subroutine subr(x, y, n)
  implicit none
  real x, y
  integer n
  x = n
  y = y*x
end subroutine subr
```

! サブルーチンの呼び出し



サブルーチン実行の流れ

．．．このようになる



- ★ サブルーチンの途中で実行を終了したいときは，return文を書く
- ★ サブルーチンの中で stop 文を実行すると，プログラムが終了する

return
stop

！ return文
！ stop文



メインプログラムとサブルーチンの位置関係

- サブルーチンとメインプログラムの順序は動作と無関係


```
program stest1
  ...
  call subr (x, y, 10)
  ...
end program stest1
subroutine subr (x, y, n)
  ...
end subroutine subr
```

=

```
subroutine subr (x, y, n)
  ...
end subroutine subr
program stest1
  ...
  call subr (x, y, 10)
  ...
end program stest1
```

- 一つのプログラムの中に入れられるメインプログラムは1個だが、サブルーチンは、名前が異なればいくつでも入れられる
- サブルーチンは記述するだけでもOK. どこからもコールされなくてもエラーにはならない
- メインプログラムとサブルーチンを別のファイルにして結合することも可能 → これが「リンク」
- 数値計算などに関する既存のサブルーチンをリンクさせることも可能 → これが「ライブラリ」

2. 引数の使い方





ローカル変数

- サブルーチン内部で宣言した変数は、他のルーチンでの宣言とは無関係、これを「ローカル変数」という
- このため、ルーチンが異なれば同じ名前の変数を宣言してもエラーにはならないし、全く異なるメモリを指定するので実体も異なる

☆ あるルーチンから他のルーチンの内部は見えない！

例えば、次のように書いても問題ないが、数値の受け渡しはできない

```
program stest1
  implicit none
  real x, y
  x = 10.0
  y = 30.0
  call subr1
end program stest1

subroutine subr1
  implicit none
  real x, y
  print *, x, y
end subroutine subr1
```

！ このxとyは、10や30ではない！



数値の受け渡し手段の一つが引数



数値の受け渡しには「引数」を使う

```
program stest2
  implicit none
  real x, y
  x = 10.0
  y = 30.0
  call subr2(x, y)
end program stest2
```

！ 引数ありサブルーチンの呼び出し

```
subroutine subr2(x, y)
  implicit none
  real x, y
  print *, x, y
end subroutine subr2
```

！ 引数ありの subroutine文

！ 引数は宣言しなければならない

- ★ 引数を使えば、数値の受け渡しができる
 - ★ 「引数」は、データの窓口であるが、あくまでもローカル変数なので、数値型や名前などはサブルーチン側で決める
 - ★ このため、call文とsubroutine文での数値の対応には気をつけなくては
- いけない

call subr2(10, 30) ！ このcall文にすると期待通りの結果にはならない！
 ！ call subr2(10.0, 30.0) と書かなければならない



プログラムを書いてみよう

3-B 変数の交換用サブルーチン

例題3-B

実変数 x と y を与えると、それに代入されている値を交換して戻るサブルーチンを作成し、実際に交換されているかどうかを確かめよ



3-B 変数の交換用サブルーチン の解答例

```
program answer3b
  implicit none
  real x, y
  x = 10; y = 50
  print *, ' x, y (before) = ', x, y
  call swap(x, y)
  print *, ' x, y (after)  = ', x, y
end program answer3b

subroutine swap(a, b)
  implicit none
  real a, b, c
  c = a
  a = b
  b = c
end subroutine swap
```

☆書き方のポイント☆

- ★ サブルーチン内で引数変数に値を代入するとcall文の変数に代入される
- ★ この機能を利用するとき、call文では必ず「変数」を指定すること

引数を使うときの規則



- 並びの順番や数は、全て一致していなければならない
- call文の数値は、対応するsubroutine文の引数と数値型が一致していなければならない
- サブルーチン内で引数変数に数値を代入すると、call文の対応する変数に値が代入される。これを「戻り値」という
- 戻り値を利用する場合には、call文の対応する引数に、必ず「同じ数値型を持つ変数」を指定しなければならない

☆ 誤った使い方 ☆

```
program answer3b
  implicit none
  real x, y
  call swap(x, x+y)
  call swap(20.0, y)
end program answer3b

subroutine swap(a, b)
  implicit none
  real a, b, c
  c = a
  a = b
  b = c
end subroutine swap
```

! このcall文はエラー（数式には代入できない）
! このcall文もエラー（定数には代入できない）



配列を引数にする場合

● 引数には，配列を指定することもできる

☆ 配列名 `a` の配列を使って，`call sub(a)` と呼び出したとき

subroutine文の引数が単一変数として宣言されているときには，
配列 `a` の先頭要素を使ってサブルーチンが動作する

```
subroutine sub(x)
  implicit none
  real x                ! 単一変数宣言
  x = 10.0
end subroutine sub
```

サブルーチン側で配列として使いたければ，引数を配列として宣言する

```
subroutine sub(x)
  implicit none
  real x(10)           ! 配列宣言
  integer i
  do i = 1, 10
    x(i) = i
  enddo
end subroutine sub
```

★ なお，コール側と要素数が一致する必要はない（見えない！）

★ 次元でさえ一致させる必要はない



プログラムを書いてみよう

3-C ベクトルの内積を計算するサブルーチン

例題3-C

整数 n と、長さ n の2個の1次元配列 a と b を引数に与えると、それぞれを n 次元ベクトルと考えたときのベクトルの内積を計算してその結果を戻り値とするサブルーチンを作成し、結果を確かめよ。

☆ n 次元ベクトルの内積☆

$A = (A_1, A_2, \dots, A_n)$, $B = (B_1, B_2, \dots, B_n)$ として

$$A \cdot B = A_1 B_1 + A_2 B_2 + \dots + A_n B_n$$

3-C ベクトルの内積を計算するサブルーチン の解答例

```
program answer3c
  implicit none
  real a(10), b(10), pab
  integer i, n
  n = 10
  do i = 1, n
    a(i) = i
    b(i) = n-i
  enddo
  call inprod(a, b, n, pab)
  print *, ' A.B = ', pab
end program answer3c

subroutine inprod(x, y, n, p)
  implicit none
  real x(10), y(10), p
  integer n, i
  p = 0
  do i = 1, n
    p = p + x(i)*y(i)
  enddo
end subroutine inprod
```

☆書き方のポイント☆

- ★ サブルーチン内の配列宣言は「10」である必要はない
- ★ この例の場合には、10 の代わりに、引数 n でも良い（整合配列）



整合配列

- Fortranでは、サブルーチンの引数配列を他の引数の中にある整数型の変数を使って、あたかもその長さの配列のように宣言することができる

☆これを「整合配列」という☆

例えば、先ほどの解答例のサブルーチンは以下のように書ける

```
subroutine inprod(x, y, n, p)
  implicit none
  real x(n), y(n), p      ! nが引数なので整合配列が可能
  integer n, i
  p = 0
  do i = 1, n
    p = p + x(i)*y(i)
  enddo
end subroutine inprod
```

★ 2次元以上の配列でも可能




整合配列はなぜ可能か？

- サブルーチンにおける「引数」は特別な存在で、call文から受け取っているのは「数値」ではなく「アドレス」である
- 引数が配列の場合には、その先頭アドレスを受け取っているだけであり、他の要素は、その先頭アドレスから数えてメモリの位置を決めている
- よって1次元配列の場合は、宣言数 n は何でも良い（* を使って $a(*)$ のように宣言することも可能）
- 下限を指定した配列は、下限を意識して引数に与えないと処理する位置が変わるので注意！
- 2次元以上の配列では、配列要素の上限がわからないといけなないので、整合配列が有用である

real a(8)	
メモリ アドレス	メモリ領域
101:	a(1)
102:	a(2)
103:	a(3)
104:	a(4)
105:	a(5)
106:	a(6)
107:	a(7)
108:	a(8)
....	



配列を引数にするときの注意



例えば、サブルーチン sub の引数を、整合配列を使って右のように宣言した場合を考える

もし、コール側のプログラムで

```
real a(10)
```

と宣言した配列をcall 文の引数に使う場合、

```
call sub(a, 10) と call sub(a(1), 10)
```

は、同じ意味になる

また

```
call sub(a(3), 8)      ! a(3)からスタートするので、要素数は 8
```

と書けば、a(3)が先頭要素の配列と見なすので、a(3)とx(1)が対応する
もし、下限を指定して、

```
real ac(0:10)
```

と宣言した配列を call 文の引数に使った場合、

```
call sub(ac, 11)      ! ac(0)からac(10)までなので、要素数は 11
```

と書けば、ac(0)が配列の先頭要素なので、ac(0)とx(1)が対応する

```
subroutine sub(x, n)  
  implicit none  
  real x(n)  
  integer n
```

☆ サブルーチンで、real x(0:n) と宣言すれば、ac(0)とx(0)が対応する



プログラムを書いてみよう

3-D 行列の和を計算するサブルーチン

例題3-D

整数 n と、 $n \times n$ の2個の2次元配列、 a と b を引数に与えると、それぞれを n 次正方行列と考えたときの行列の和を計算して、別の2次元配列に代入して戻るサブルーチンを作成し、結果を確かめよ。

3-D 行列の和を計算するサブルーチン の解答例


```
program answer3d
  implicit none
  real a(5,5), b(5,5), c(5,5)
  integer i, j, n
  n = 5
  do j = 1, n
    do i = 1, n
      a(i,j) = i*j
      b(i,j) = 2*i-j
    enddo
  enddo
  call matrix_sum(a, b, n, c)
  do i = 1, n
    print *, i, (c(i,j), j=1, n)
  enddo
end program answer3d

subroutine matrix_sum(x, y, n, z)
  implicit none
  real x(n,n), y(n,n), z(n,n)
  integer n, i, j
  do j = 1, n
    do i = 1, n
      z(i,j) = x(i,j) + y(i,j)
    enddo
  enddo
end subroutine matrix_sum
```

☆書き方のポイント☆

- ★ 2次元配列を受け渡すときには、整合配列を利用すると便利
- ★ この例では、第1要素と第2要素の宣言がどちらも n であるが、別の引数 m を加えて、real x(m,n) のように宣言することも可能
- ★ 配列の出力に do型出力を使用している（第2部 付録2C参照）

3. 関数副プログラム





関数副プログラムとは

- $y=3*\text{fun}(x)$ のように数式中で使える自前の関数を作ることにもできる
これを「関数副プログラム」という
- 関数副プログラムは、サブルーチンとほぼ同じ構造だが「関数名」が
1 個の変数であり、その変数が関数値として数式中で使えるものである
- このため、使うときには、関数名の型宣言をしなければならない

☆ 関数副プログラムとサブルーチンの相違点 ☆

- ★ 関数副プログラムは、function文と end function文の間に書く
- ★ 関数名を変数として宣言し、それに計算結果（関数値）を代入しなければならない

例えば、

```
function square(x)
  implicit none
  real square, x
  square = x*x
end function square
```

```
! function文
!
! 関数名と引数の型宣言
! 関数名に値を代入する
! end function文
```

のように書く



プログラムを書いてみよう

3-E ベクトルの長さを計算する関数

例題3-E

整数 n と、長さ n の1次元配列 a を引数に与えると、 a の要素を n 次元ベクトルの成分と考えたときのベクトルの長さを関数値とする関数副プログラムを作成し、それを使って適当な成分値を代入した2個の n 次元ベクトルの長さの合計を計算して出力するプログラムを作成せよ。

☆ n 次元ベクトルの長さ☆

$\mathbf{A} = (A_1, A_2, \dots, A_n)$ として

$$|\mathbf{A}| = \sqrt{A_1^2 + A_2^2 + \dots + A_n^2}$$



3-E ベクトルの長さを計算する関数 の解答例

```
program answer3e
  implicit none
  real a(10), b(10), sum, length
  integer i, imax
  imax = 10
  do i = 1, imax
    a(i) = 2.5*i + 0.5
    b(i) = 1.25*i - 10
  enddo
  sum = length(a, imax) + length(b, imax)
  print *, 'Length(a)+Length(b) = ', sum
end program answer3e

function length(x, n)
  implicit none
  real length, x(n), x1
  integer n, i
  x1 = 0
  do i = 1, n
    x1 = x1 + x(i)**2
  enddo
  length = sqrt(x1)
end function length
```

☆書き方のポイント☆

- ★ 関数を利用するメインプログラムでは、関数の名前を型宣言する
- ★ 関数副プログラム中でも関数名を宣言し、その関数名を変数として数値を代入すれば、その数値が関数値となる



result句を使った関数副プログラム

🌐 result句を利用すれば関数名に代入する形にしなくても良い

例えば,

```
function square(x) result(y)  ! result句を付加
  implicit none
  real x,y                    ! 関数名ではなく, result句で指定した変数を型宣言
  y = x*x                     ! result句で指定した変数に関数値を代入する
end function square
```

のように書くことができる

利用するときは「関数名」を型宣言しなければならないのは同じ

```
program ftest1
  implicit none
  real x,y,square              ! 使用する関数名を型宣言する
  x = 5.2
  y = 3.0*square(x+1.0) + 50.5
  print *,x,y
end program ftest1
```

4. グローバル変数



```
module mod1  
  real xax, yax  
  .....  
end module mod1  
モジュール
```

```
program prog1  
  .....  
  xax = 10  
  call sub1  
  .....  
end program prog1  
メインプログラム
```

```
subroutine sub1  
  .....  
  yax = xax**2  
  .....  
end subroutine sub1  
サブルーチン
```



グローバル変数を使って変数を共有する

- 大きなプログラムになると引数だけでデータを受け渡すのは不便
- 複数のルーチンで値を共有したいときには「グローバル変数」を利用すると便利
- Fortranではモジュールを使ってグローバル変数を利用する
- モジュールにはより高度な使い方もある（本講座では説明を省略）

モジュールの構造

```
module モジュール名  
  宣言文1  
  宣言文2  
  .....  
end module モジュール名
```

- ★ モジュールはメインプログラムやサブルーチンとは独立しているので外部に書く
- ★ 利用する全てのルーチンより前に書く
- ★ モジュールの中から他のモジュールを利用することも可能



モジュールの利用手段： use 文

use文

use モジュール名

モジュールはuse文で利用を宣言する


```
module global          ! モジュールは前に書く
  real xaxis, yaxis
end module global

program stest4
  use global            ! implicitより前でuse宣言
  implicit none
  xaxis = 5.0
  yaxis = 100.0
  call subr4
  print *, xaxis, yaxis
end program stest4

subroutine subr4
  use global            ! implicitより前でuse宣言
  implicit none
  print *, xaxis, yaxis
  yaxis = 25.0
end subroutine subr4
```

- グローバル変数を利用するルーチン内で、use文を使ってモジュールの利用を宣言する
- 逆に use文で宣言しなければ、そのモジュールで宣言されたグローバル変数は使えない（同じ名前をローカル宣言しても良いが、別物）
- use文はimplicit noneよりも前に書く

モジュール利用のイメージ



★ 外部の共有メモリ領域を参照するところがポイント



プログラムを書いてみよう

3-F グローバル変数を利用した 行列の和を計算するサブルーチン

例題3-F

例題3-Dの n 次正方行列の和を計算するプログラムを，特定の配列とその次元数をグローバル変数として用意したモジュールを利用したプログラムに書き換えよ．



3-F グローバル変数を利用したサブルーチン の解答例

```
module matrix
  real amat(5,5), bmat(5,5), cmat(5,5)
  integer nmax
end module matrix
```

P-1

```
program answer3f
  implicit none
  call initialize
  call matrix_plus
  call output
end program answer3f
```

```
subroutine initialize
  use matrix
  implicit none
  integer i, j
  nmax = 5
  do j = 1, nmax
    do i = 1, nmax
      amat(i,j) = i*j
      bmat(i,j) = 2*i-j
    enddo
  enddo
```

```
    enddo
  enddo
end subroutine initialize
```

P-2

```
subroutine matrix_plus
  use matrix
  implicit none
  integer i, j
  do j = 1, nmax
    do i = 1, nmax
      cmat(i,j) = amat(i,j) + bmat(i,j)
    enddo
  enddo
end subroutine matrix_plus
```

```
subroutine output
  use matrix
  implicit none
  integer i, j
  do i = 1, nmax
    print *, i, (cmat(i,j), j=1,nmax)
  enddo
end subroutine output
```

☆書き方のポイント☆

- ★ 特定のデータの計算のみで構成されている場合にはグローバル変数を使うのが有効
- ★ メインプログラムは各サブルーチンのコールだけにするとメンテナンスがしやすい
 - 計算機シミュレーションプログラムを構成するときに考慮する
- ★ モジュールで宣言するグローバル変数は意味のある長い名前にする
- ★ 次元数のような共通変数もグローバル変数にすると使いやすい
- ★ do 文のカウンタ変数などは、ローカル変数にする



グローバル変数を利用するときの注意

- むやみやたらにグローバル変数にしない
- 用途に応じて変数を分類し，それぞれのモジュールを作る
- グローバルであることを意識した意味のある長めの変数名にする

必要とするグローバル変数が少ないときは，宣言を限定することもできる

```
use モジュール名, only : 変数 1, 変数 2, . . .
```

例えば，xaxisとyaxisが宣言されているモジュールを考える

```
module global  
  real xaxis, yaxis  
end module global
```

このモジュールglobalを使うときに必要なのが yaxis のみの場合には，

```
use global, only : yaxis
```

と書けば， yaxisのみ使える

xaxisは使えないし， 別のローカル変数として宣言してもかまわない

第3部 終了





Fortranには便利な書き方がまだまだあります

- 便利なデータ入力方法
- 文字列の処理
- 配列計算
- ビット操作関数や乱数発生

数値計算法も含めて勉強するなら
拙著もあります！



- ★ 東北大学より、もう少し詳しい内容が入った解説書をダウンロードすることができます
- ★ より高度に使いこなしてみたいと思ったら他の解説書を読んで勉強して下さい

Fortranハンドブック
田口俊弘 著
技術評論社

東北大学サイバーサイエンスセンター講習会

第8回 Fortran入門

これで今回の講習会は終了です

お疲れ様でした！



付録：知っておくと便利な文法 3





3A. 拡張宣言文

これまでの型宣言文では、数値型の指定と変数名（配列）のみ記述

型指定 変数1, 変数2, ...

拡張宣言文では、属性や初期数値を代入することができる

型指定 [, 属性1, 属性2, ...] :: 変数1 [=数値1], 変数2 [=数値2], ...

- ★ 拡張宣言文には、連続した2個のコロン「::」が必要である
- ★ なお、この文で [...] は省略可能という意味で使っているだけなので、“[”と“]”はプログラムには書かないようにして下さい

たとえば、

```
integer :: imax=10, jmax=100  
real    :: xx=1.0, yy=2.0
```

のように宣言と同時に数値代入しておくと、この値でプログラムが開始する



拡張宣言文における数値代入の意味

- 拡張宣言文での数値代入は最初の1回だけである
- 実行文中で変更したら変更後の値が残る

たとえば、次の例ではサブルーチン subr1をコールすることにより出力が異なる

```
program stest1
  implicit none
  call subr1
  call subr1
  call subr1
end program stest1

subroutine subr1
  implicit none
  integer :: n = 1
  print *, n
  n = n + 1
end subroutine subr1
```

！ コールするたびに n は増加する

★ すなわち、宣言文での代入は、プログラム開始後の1回だけ

ここで、サブルーチンのローカル変数に値が保持されていることに注目！



一時メモリ領域と固定メモリ領域

- 実は、サブルーチン内で宣言した変数はサブルーチン終了後に消滅すると考えなければならない → 一時メモリ領域

たとえば、次のプログラムを考えてみよう

```
program stest2
  implicit none
  call subr2(1)
  call subr2(2)
end program stest2

subroutine subr2(n)
  implicit none
  integer n
  real x, y
  if (n == 2) print *, x, y
  x = 10.0
  y = 100.0
end subroutine subr2
```

！ この出力値は不定

この場合、2回目のコールで10と100が出力されるという保証はない！

☆ 数値代入しておくと、消滅しない → 固定メモリ領域



save属性を付加すると固定メモリになる



数値代入しないで固定メモリ領域に所属させるには， save属性を付ける

```
integer, save :: 変数1, 変数2, ...
```

★ この変数に代入した値は， サブルーチン終了後も保持される

先ほどのプログラムは，

```
program stest2
  implicit none
  call subr2(1)
  call subr2(2)
end program stest2

subroutine subr2(n)
  implicit none
  integer n
  real, save :: x, y      ! save属性を付加
  if (n == 2) print *, x, y ! この出力値は10.0と100.0
  x = 10.0
  y = 100.0
end subroutine subr2
```

のように修正すれば， 期待通りの結果になる



3B. parameter 変数

- 配列宣言は「定数」で範囲を指定しなければならないが、これは、プログラムの開始時にメモリを確保しなければならないためである
- 配列の数が増えると修正が面倒である

たとえば,

```
real a(100), b(100)           ! 100が2カ所
integer i
do i = 1, 100                 ! 100が1カ所
  a(i) = i*i
  b(i) = a(i)**2
enddo
```

100を変更したいときは、3カ所修正が必要！

parameter属性を付加したparameter変数を使えば、変数で宣言できる

```
integer, parameter :: imax=100 ! 修正はここだけでよい
real a(imax), b(imax)         ! imax で宣言ができる
integer i
do i = 1, imax
  a(i) = i*i
  b(i) = a(i)**2
enddo
```



parameter 変数は置き換えるだけ



parameter変数を使って、別のparameter変数を定義したり、
下限指定や計算式を使った指定も可能

たとえば、

```
integer, parameter :: imax=100, imax2=imax**2  
real a(imax-1), b(0:imax*2), c(-imax2:imax2)
```

は、以下のプログラムと等価である

```
integer, parameter :: imax=100, imax2=10000  
real a(99), b(0:200), c(-10000:10000)
```

☆ ただし、parameter変数は、コンパイル時に単語を数字に置き換えているだけなので、実行文で変更することはできない

たとえば、次のように書くとコンパイルエラーになる

```
integer, parameter :: imax=100  
imax = 5                ! これはエラー
```

なぜなら、プログラムの際には次のように書いたことになるから

```
integer, parameter :: imax=100  
100 = 5                  ! このプログラムを書いたことになるので
```



parameter 変数を活用しよう

parameter変数をモジュールで用意すれば、長いプログラムで便利

```
module global_param
  integer, parameter :: imax=300, jmax=200
end module global_param
```

```
module mod_array
  use global_param
  real abc(imax, jmax), cd(jmax*2-1)
end module mod_array
```

！ モジュール内でもuse宣言可能
！ グローバル配列宣言

```
program mtest1
  use mod_array
  implicit none
  integer km(imax)
  .....
```

！ use global_paramは不要
！ ローカル配列宣言

物理定数などを用意しておくのも便利（CODATA 2018推奨値）

```
module physical_constants
  real, parameter :: clight =299792458.0,      hplanck=6.62607015e-34
  real, parameter :: echarge=1.602176634e-19,  kboltz =1.380649e-23
  real, parameter :: emass =9.1093837015e-31,  pmass =1.67262192369e-27
end module physical_constants
```



3C. 配列の動的割り付け

動的割り付け：実行時に必要に応じたメモリ領域を確保すること

Fortranでの動的割り付けには2つの方法がある

動的割り付けその1：引数整数で宣言する

たとえば,

```
subroutine memory1(a,m)
  implicit none
  real a(m), b(m)
  integer m, i
  do i = 1, m
    b(i) = 2.5*i - 3
    a(i) = b(i)**3
  enddo
end subroutine memory1
```

！ 引数 m を使ってローカル配列を宣言する

と書くことができる

- ★ このサブルーチンで, $a(m)$ は整合配列
- ★ $b(m)$ はこのサブルーチン内で割り付けられたメモリ領域
- ★ あまり大きなサイズを割り付けると, エラーになることもある

配列の動的割り付け

動的割り付けその2 : allocatable宣言とallocate文

- 引数による動的割り付けは、サブルーチンでしか使えない
- メインプログラムで使う配列やグローバル配列を割り付けたいときには、allocatable 宣言と allocate文を使って、より明示的に割り付ける

割り付け手順 その1

割り付ける予定の配列は、allocatable属性を付けて宣言する

```
real, allocatable :: ab(:), z2(:, :)    ! ab は 1次元配列, z2 は 2次元配列
integer, allocatable :: km1(:, :, :)    ! km1 は 3次元配列
```

- ★ その際、配列要素数は不要だが、次元は後から決められないので、コロンの数で指定しておく

割り付け手順 その2

必要に応じて allocate文で割り付けを実行する

```
allocate ( ab(100), z2(0:m, -m:m), km1(k-1, 2*k, 5) )    ! 両端のかっこは必須！
```

- ★ このように、変数や数式を使って割り付けることも可能
- ★ 数値型に関係なく 1回の allocate文で複数の配列を割り付けられる



動的割り付け配列の解放と確認

allocatable宣言とallocate文による割り付けの特長

- ★ メインプログラムでも可能
- ★ モジュールの配列もallocatable宣言ができる
- ★ 必要に応じてメモリを解放することができる

割り付けたメモリ領域の解放：deallocate文

```
deallocate ( ab, km1 )
```

 ! 両端のかっこは必須！

- ★ deallocate文では配列名のみ記述する
- ★ 数値型に関係なく 1 文で複数の配列を解放できる

割り付けられているかどうかの確認：allocated関数（論理型）

たとえば、配列 a が割り付けられていないときのみ割り付けたいときは、

```
if ( .not. allocated(a) ) allocate ( a(n) )
```

のように書くことができる（論理型は 第2部 付録2B参照）



例えば例題2-Cの解答は以下のように書き換えられる

```
program answer2c_alloc
  implicit none
  real, allocatable :: a(:)
  real sum
  integer i, imax
  imax = 30
  allocate ( a(imax) )
  a(1) = 1;    a(2) = 1
  do i = 3, imax
    a(i) = a(i-1) + a(i-2)
  enddo
  sum = 0
  do i = 1, imax
    sum = sum + a(i)
  enddo
  deallocate ( a )
  print *, ' Average = ', sum/imax
end program answer2c_alloc
```

☆書き方のポイント☆

- ★ このプログラムなら、原理的には imax がいくつでも動作可能
- ★ プログラムが終了すればメモリは解放されるので、このプログラムでは deallocate 文が無くても問題はない